# 2ExtensionModule

```
> restart;
with(Involutive): with(homalg):
`homalg/default`:=`Involutive/homalg`;
var:=[x,y,z];
```

$$homalg/default := Involutive/homalg$$

$$var := [x, y, z]$$

```
> RP:=`Involutive/homalg`;
```

$$RP := Involutive/homalg$$

We start with a module $E$ and construct a 2-fold extension

$$0 \leftarrow M \xleftarrow{\pi} G_0 \xleftarrow{\varphi} G_1 \xleftarrow{\iota} N \leftarrow 0$$

within it. When we consider this extension, the module we are looking for definitely exists.

```
> E:=Cokernel([[x^4 + y^3*z,y^4,z^4],[y,z,x^5]],var);
```

$$E := [[[1, 0, 0] = [1, 0, 0], [0, 1, 0] = [0, 1, 0], [0, 0, 1] = [0, 0, 1]], [[x^4 + y^3\, z, y^4, z^4], [y, z, x^5]], \text{“Presentation”}, 3 + 9\,s + 18\,s^2 + 30\,s^3 + 44\,s^4 + s^5\,(\frac{44}{1-s} + \frac{14}{(1-s)^2} + \frac{1}{(1-s)^3}), [44, 14, 1]]$$

Add relations to $E$ to get a factor module $M$:

```
> _phi:=[[x^2,y^2,z^2],[x^3,y^3,z^3]];
```

$$\_\varphi := [[x^2, y^2, z^2], [x^3, y^3, z^3]]$$

```
> M:=Cokernel(_phi,E,var):
> pi:=CokernelEpi(_phi,E,var);
```

$$\pi := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> G1:=Kernel(E,pi,M,var);
```

$$G1 := [[[1, 0] = [x^2, y^2, z^2], [0, 1] = [0, -y^3 + x\,y^2, x\,z^2 - z^3]], [[y^5\,z^2\,x - y^5\,z^3 - 2\,y^3\,z^4\,x + y^4\,z^4 + y^3\,z^5 - z\,x^5\,y^6 + z\,x^6\,y^5 + x^{10}\,y^2 - x^9\,y^3 - x^5\,z^3 + z^4\,x^4, -y^5\,z^2 + 2\,y^3\,z^4 - z^5\,x^2 - y^5\,z\,x^5 - x^9\,y^2 + x^7\,y^4 + z^3\,x^4]], \text{“Presentation”}, 2 + 6\,s + 12\,s^2 + 20\,s^3 + 30\,s^4 + 42\,s^5 + 56\,s^6 + 72\,s^7 + 90\,s^8 + 110\,s^9 + 132\,s^{10} + 156\,s^{11} + s^{12}\,(\frac{156}{1-s} + \frac{24}{(1-s)^2} + \frac{1}{(1-s)^3}), [156, 24, 1]]$$

```
> iota_:=KernelEmb(E,pi,M,var);
```

$$iota\_ := \begin{bmatrix} x^2 & y^2 & z^2 \\ 0 & -y^3 + x\,y^2 & x\,z^2 - z^3 \end{bmatrix}$$

Add fewer relations to $E$ to get a factor module $G_0$ that itself has $M$ as a factor:

```
> G0:=Cokernel([_phi[2]],E,var);
```

$$G0 := [[[1, 0, 0] = [1, 0, 0], [0, 1, 0] = [0, 1, 0], [0, 0, 1] = [0, 0, 1]], [[x^3, y^3, z^3], [-y^3\,z, x\,y^3 - y^4, x\,z^3 - z^4], [y, z, x^5]], \text{“Presentation”}, 3 + 9\,s + 18\,s^2 + 29\,s^3 + 41\,s^4 + s^5\,(\frac{41}{1-s} + \frac{12}{(1-s)^2}), [41, 12, 0]]$$

```
> lambda:=CokernelEpi([_phi[2]],E,var);
```

$$\lambda := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> phi:=ComposeMaps(iota_,lambda,G0,var);
```

$$\varphi := \begin{bmatrix} x^2 & y^2 & z^2 \\ 0 & -y^3 + x\,y^2 & x\,z^2 - z^3 \end{bmatrix}$$

```
> N := Kernel(G1, phi, G0, var);
```

$$N := [[1 = [x^3,\, y^3,\, z^3]],\, [0],\, \text{“Presentation”},\, \frac{1}{(1-s)^3},\, [0,\, 0,\, 1]]$$

```
> iota := KernelEmb(G1, phi, G0, var);
```

$$\iota := \begin{bmatrix} x & -1 \end{bmatrix}$$

The resulting 2-extension is

```
> TwoExtension:=N,iota,G1,phi,G0,pi,M:
> IsExactCoseq([TwoExtension],var);
```

$$\mathit{true}$$

(extensions have to be given in reverse order, i.e. the arrows point to the right), it is trivial in $\mathrm{YExt}^2(M, N)$ as expected.

```
> CocycleOfExtension(TwoExtension,var,"return_abstract_generator");
```

$$0 = \begin{bmatrix} y^3\,z^3\,x - y^4\,z^2\,x - z^4\,x^3 + z^3\,x^4 + x^8\,y^3 - x^9\,y^2 \end{bmatrix}$$

We can factorize the extension into the two 1-extensions and then compute the corresponding 1-cocycles $\eta_L^M$ and $\eta_N^L$:

```
> OneExtensions:=`homalg/FactorizeExtension`(TwoExtension,var):
> etaML := CocycleOfExtension(op(OneExtensions[2]),var);
```

$$\mathit{etaML} := \begin{bmatrix} 1 \\ x \\ x\,y \\ 0 \end{bmatrix}$$

```
> etaLN := CocycleOfExtension(op(OneExtensions[1]),var);
```

$$\mathit{etaLN} := \begin{bmatrix} -x^7\,y^4 + z^5\,x^2 - 2\,y^3\,z^4 - z^3\,x^4 + y^5\,z^2 + y^5\,z\,x^5 + x^9\,y^2 \end{bmatrix}$$

The module $L$ is $G_0/N$, it is the last module of the first 1-extension. The normalization of the modules is needed for computational reasons.

```
> L :=
NormalizeInput(ResolutionOfModule(OneExtensions[1][-1],var,"TRUNCATE"=
0)[1]);
```

$$L := [[y^5\,z^3 - y^4\,z^4 - y^3\,z^5 + z\,x^5\,y^6 - x^8\,y^4 + x^9\,y^3 - z^4\,x^4 + z^5\,x^3]]$$

```
> M := NormalizeInput(ResolutionOfModule(M,var,"TRUNCATE"=0)[1]);
```

$$M := [[x^2,\, y^2,\, z^2],\, [0,\, -y^3 + x\,y^2,\, x\,z^2 - z^3],\, [y^3\,z,\, 0,\, y\,x\,z^2 - y\,z^3 - x\,z^3 + z^4],\, [y,\, z,\, x^5]]$$

```
> N := NormalizeInput(ResolutionOfModule(N,var,"TRUNCATE"=0)[1]);
```

$$N := [[0]]$$

The cocycles are both not trivial:

```
> Ext(1,M,L,var,"var_to_assign_proc_to_express_generators_abstractly"='
abstractly1_ML'):
> abstractly1_ML(etaML);
```

$$1$$

```
> Ext(1,L,N,var,"var_to_assign_proc_to_express_generators_abstractly"='
abstractly1_LN');
```

$$[[1 = \begin{bmatrix} 1 \end{bmatrix}],\, [y^5\,z^3 - y^4\,z^4 - y^3\,z^5 + z\,x^5\,y^6 - x^8\,y^4 + x^9\,y^3 - z^4\,x^4 + z^5\,x^3],\, \text{“Presentation”}, 1 \\ + 3\,s + 6\,s^2 + 10\,s^3 + 15\,s^4 + 21\,s^5 + 28\,s^6 + 36\,s^7 + 45\,s^8 + 55\,s^9 + 66\,s^{10} + 78\,s^{11} \\ + s^{12}\,(\frac{78}{1-s} + \frac{12}{(1-s)^2}),\, [78,\, 12,\, 0]]$$

```
> abstractly1_LN(etaLN);
```

$$-x^7\,y^4+z^5\,x^2-2\,y^3\,z^4-z^3\,x^4+y^5\,z^2+y^5\,z\,x^5+x^9\,y^2$$

Equivalently, another procedure to verify that the 2-extension is trivial computes the YONEDA product of the 1-cocycles.

```
> YonedaProductOfCocycles(M,etaML,1,L,etaLN,1,N,var,"return_abstract_ge
nerator");
```

$$0=\left[\ -y^5\,z^2+2\,y^3\,z^4-z^5\,x^2-y^5\,z\,x^5-x^9\,y^2+x^7\,y^4+z^3\,x^4\ \right]$$

so we know that a 2-extension module exists. The procedure procedure `A2ExtensionModule` finds a particular solution to the linear system:

```
> eta := A2ExtensionModule(M,etaML,L,etaLN,N,var);
```

$$\eta:=\begin{bmatrix}0\\1\\x+y\\0\end{bmatrix}$$

So all solutions are this $\eta$ plus any 1-cocycle from the first extension group of 1-cocycles $\mathrm{Ext}^1(\eta_L^M,\eta_N^L)$.

```
> Ext1_MN := Ext(1,M,N,var);
```

$$\mathit{Ext1\_MN}:=\left[\left[\left[1=\begin{bmatrix}y^2\\-y^3+x\,y^2\\0\\z\end{bmatrix}\right]\right],\ [1],\ \text{“Presentation”},\ 0,\ [0,\,0,\,0]\right]$$

We see that in this case $\mathrm{Ext}^1(M,N)=0$ and therefore $\mathrm{Ext}^1(\eta_L^M,\eta_N^L)=0$. So $\eta$ is in fact the only valid solution and $(\eta_L^M,\eta_N^L)$ is rigid.

Now we can write down the resulting relation matrix `E`:

```
> zero_21 := matrix(RP[NumberOfRows](L),RP[NumberOfGenerators](M),0);
zero_31 := matrix(RP[NumberOfRows](N),RP[NumberOfGenerators](M),0);
zero_32 := matrix(RP[NumberOfRows](N),RP[NumberOfGenerators](L),0);
```

$$\mathit{zero\_21}:=\left[\ 0\quad 0\quad 0\ \right]$$

$$\mathit{zero\_31}:=\left[\ 0\quad 0\quad 0\ \right]$$

$$\mathit{zero\_32}:=\left[\ 0\ \right]$$

```
> EE:=linalg[blockmatrix](3,3,[
M,etaML,eta,
zero_21,L,etaLN,
zero_31,zero_32,N]);
```

$\mathit{EE}:=$

$[x^2\,,\,y^2\,,\,z^2\,,\,1\,,\,0]$

$[0\,,\,-y^3+x\,y^2\,,\,x\,z^2-z^3\,,\,x\,,\,1]$

$[y^3\,z\,,\,0\,,\,y\,x\,z^2-y\,z^3-x\,z^3+z^4\,,\,x\,y\,,\,x+y]$

$[y\,,\,z\,,\,x^5\,,\,0\,,\,0]$

$[0\,,\,0\,,\,0\,,\,y^5\,z^3-y^4\,z^4-y^3\,z^5+z\,x^5\,y^6-x^8\,y^4+x^9\,y^3-z^4\,x^4+z^5\,x^3\,,$
$-x^7\,y^4+z^5\,x^2-2\,y^3\,z^4-z^3\,x^4+y^5\,z^2+y^5\,z\,x^5+x^9\,y^2]$

$[0\,,\,0\,,\,0\,,\,0\,,\,0]$

```
> Cokernel(EE,var);
```

$$[[[1,\,0,\,0]=[1,\,0,\,0,\,0,\,0],\,[0,\,1,\,0]=[0,\,1,\,0,\,0,\,0],\,[0,\,0,\,1]=[0,\,0,\,1,\,0,\,0]],$$
$$[[x^4+y^3\,z,\,y^4,\,z^4],\,[y,\,z,\,x^5]],\ \text{“Presentation”},$$
$$3+9\,s+18\,s^2+30\,s^3+44\,s^4+s^5\left(\frac{44}{1-s}+\frac{14}{(1-s)^2}+\frac{1}{(1-s)^3}\right),\ [44,\,14,\,1]]$$

So the resulting module is exactly the module $E$ we started with.

Author: Barbara Bremer

| | |
|---|---|
| Date: | 2008-02-27 |
| Last modified: | 2008-02-27 16:30 |

## References


[BB] Mohamed Barakat and Barbara Bremer, *Higher Extension Modules and the Yoneda Product*, submitted http://wwwb.math.rwth-aachen.de/homalg).

[BCG+03] Y. A. Blinkov, C. F. Cid, V. P. Gerdt, W. Plesken, and D. Robertz, *The* MAPLE *Package* Janet*:* I. *Polynomial Systems.* II. *Linear Partial Differential Equations*, Proc. 6th Int. Workshop on Computer Algebra in Scientific Computing, Passau, Germany, 2003, (http://wwwb.math.rwth-aachen.de/Janet), pp. 31–40 and 41–54.

[BR] Mohamed Barakat and Daniel Robertz, `homalg` – *A meta-package for homological algebra*, accepted for publication in Journal of Algebra and its Applications. (arXiv:math.AC/0701146 and http://wwwb.math.rwth-aachen.de/homalg).

[BR08] ———, `homalg` *project*, 2003-2008, (http://wwwb.math.rwth-aachen.de/homalg).



Lehrstuhl B für Mathematik, RWTH-Aachen University, 52062 Germany

*E-mail address*: barbara.bremer@rwth-aachen.de